Tema 10
Funciones de varias variables.
Límites, continuidad y diferenciabilidad.

Objetivos:

1. Definir funciones de varias variables con wxMaxima. Calcular con funciones de varias variables.
2. Representar funciones de dos variables gráficamente. Practicar con las diferentes opciones.
3. Calcular límites de funciones reales de varias variables.
4. Estudiar la continuidad de funciones de varias variables.
5. Calcular derivadas parciales de funciones reales de varias variables
6. Calcular matrices jacobianas y estudiar las propiedades.
7. Calcular derivadas de orden superior de funciones de varias variables.

Contenidos:

10-1. Funciones de varias variables.
10-2. Representación gráfica de funciones reales de dos variables.
10-3. Límites de funciones reales de dos variables. Continuidad.
10-4. Derivadas parciales. Derivadas direccionales. Derivadas de orden superior.
10-5. Matriz jacobiana. Diferenciabilidad. La regla de la cadena.

Referencias

AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.

APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.

AP86 APOSTOL, T.M. (1986)
Análisis Matemático

BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima

ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.

RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima

RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería

RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.

VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

10-1.- Funciones de varias variables.

Los contenidos de este apartado se desarrollan en el archivo **Tema_10-1.wxm**.

10.1.1 Funciones reales de varias variables.

Se llama función real de varias variables o campo escalar, cualquier terna (A, B, f) en la que $A \subset \mathbb{R}^n, B \subset \mathbb{R}$ y f es una correspondencia entre los dos conjuntos, tal que para todo elemento $x \in A$ existe un único elemento asignado $y \in B$. Para expresar una función real de varias variables es habitual escribir una ecuación de la forma $y = f(x_1, x_2, ..., x_n),\ x \in A \subset \mathbb{R}^n$.

Notación:

- Si n = 2, se acostumbra a escribir $x_1 = x, x_2 = y$.
- Si n = 3, se acostumbra a escribir $x_1 = x, x_2 = y, x_3 = z$.

Para calcular el dominio o campo de existencia de una función real de varias variables, hay que establecer el mayor subconjunto en el que tiene sentido la expresión que determina el valor de la imagen.

Veamos cómo se definen las funciones reales de varias variables con wxMaxima mediante algunos ejemplos.

Ejemplo 10.1.1.

(a) Se considera la función definida por: $f(x, y) = 2(x-1)^2 + 2(y-1)^2$

Para definirla se sigue una sintaxis muy similar al caso de funciones reales de variable real:

```
(%i1) f1(x,y):=2*(x-1)^2+2*(y-1)^2;
```

(%o1) $f1(x,y) := 2\,(x-1)^2 + 2\,(y-1)^2$

Para calcular el valor de la función en un punto, sencillamente:

```
(%i2) 'f1(2,3)=f1(2,3);
```

(%o2) $f1(2,3) = 10$

(b) El volumen de un cilindro recto circular es un número real positivo que depende del radio de la base (r) y de la altura (h) y, por lo tanto, se puede expresar mediante una función real de dos variables: $V(r,h) = \pi r^2 h,\quad r,h > 0$.

Entonces:

```
(%i3) V(r,h):=%pi*r^2*h;
```

$$(\%o3)\ V(r,h):=\pi\, r^2\, h$$

```
(%i4) 'V(5,7)=V(5,7);
```

$$(\%o4)\ V(5,7)=175\,\pi$$

(c) Función real de tres variables

```
(%i5) f2(x,y,z):=(x+y)^2+exp(z);
      'f2(1,2,3)=f2(1,2,3);
```

$$(\%o5)\ f2(x,y,z):=(x+y)^2+\exp(z)$$

$$(\%o6)\ f2(1,2,3)=\%e^3+9$$

(d) Función real de cuatro variables:

```
(%i7) f3(x,y,u,v):=x^2+x^3*y-2*u^2+exp(v);
      'f3(1,-1,1,2)=f3(1,-1,1,2);
```

$$(\%o7)\ f3(x,y,u,v):=x^2+x^3\,y+(-2)\,u^2+\exp(v)$$

$$(\%o8)\ f3(1,-1,1,2)=\%e^2-2$$

10.1.2.- Operaciones con funciones reales de varias variables

Se pueden llevar a cabo operaciones con funciones reales de varias variables. Consideremos por ejemplo las funciones reales de dos variables:

$$f_1(x,y)=2x^2+y^2+1;\quad f_2(x,y)=x^2\sin(x+y)$$

Definición de las funciones con wxMaxima:

```
(%i1) f1(x,y):=2*x^2+y^2+1;     f2(x,y):=x^2*sin(x+y);
```

$$(\%o1)\ f1(x,y):=2\,x^2+y^2+1$$

$$(\%o2)\ f2(x,y):=x^2\sin(x+y)$$

Suma:

```
(%i3) '(f1(x,y)+f2(x,y))=f1(x,y)+f2(x,y);
```

$$(\%o3)\ f2(x,y)+f1(x,y)=x^2\sin(y+x)+y^2+2\,x^2+1$$

Producto por un escalar:

(%i4) '(a*f1(x,y))=a*f1(x,y);

(%o4) $a\, f1(x,y)=a\,(y^2+2\,x^2+1)$

(%i5) expand(%);

(%o5) $a\, f1(x,y)=a\,y^2+2\,a\,x^2+a$

Producto:

(%i7) '(f1(x,y)*f2(x,y))=f1(x,y)*f2(x,y);

(%o7) $f1(x,y)\,f2(x,y)=x^2\,(y^2+2\,x^2+1)\sin(y+x)$

(%i8) expand(%);

(%o8) $f1(x,y)\,f2(x,y)=x^2\,y^2\sin(y+x)+2\,x^4\sin(y+x)+x^2\sin(y+x)$

Cociente:

(%i9) '(f1(x,y)/f2(x,y))=f1(x,y)/f2(x,y);

(%o9) $\dfrac{f1(x,y)}{f2(x,y)}=\dfrac{y^2+2\,x^2+1}{x^2\sin(y+x)}$

10.1.3.- Funciones vectoriales de varias variables.

Se llama <u>función vectorial de varias variables o campo vectorial</u>, cualquier terna (A,B,f) en la que $A\subset\mathbb{R}^n, B\subset\mathbb{R}^m$ y f es una correspondencia entre los dos conjuntos, tal que para todo elemento $x\in A$ existe un único elemento asignado $y\in B$.
Para expresar una función vectorial de varias variables es habitual escribir un vector definido mediante las componentes de la función, que son funciones reales de varias variables. Así, pues:

$$f(x_1,x_2,\dots,x_n)=\left(f_1(x_1,x_2,\dots,x_n),\dots,f_m(x_1,x_2,\dots,x_n)\right),\ x\in A\subset\mathbb{R}^n$$

Veamos la definición con wxMaxima mediante un ejemplo:

(%i1) f1(x,y,z):=x^2+y-sin(z); f2(x,y,z):=z*exp(x+y);
f(x,y,z):=[f1(x,y,z),f2(x,y,z)];

(%o1) $f1(x,y,z):=x^2+y-\sin(z)$

(%o2) $f2(x,y,z):=z\exp(x+y)$

(%o3) $f(x,y,z):=[f1(x,y,z),f2(x,y,z)]$

(%i4) 'f(1,2,-1)=f(1,2,-1);

(%o4) $f(1,2,-1)=[\sin(1)+3,-\%e^3]$

10.1.4.- Operaciones con funciones vectoriales de varias variables

Si se consideran dos funciones:

$$f : A_1 \subset \mathbb{R}^n \to \mathbb{R}^m, \quad g : A_2 \subset \mathbb{R}^n \to \mathbb{R}^m$$

se pueden llevar a cabo componente a componente las operaciones suma y producto por un escalar real.

Suma:

```
(%i1) f1(x,y,z):=x^2+y-sin(z);        f2(x,y,z):=z*exp(x+y);
      f(x,y,z):=[f1(x,y,z),f2(x,y,z)];
```

(%o1) $f1(x,y,z):=x^2+y-\sin(z)$

(%o2) $f2(x,y,z):=z\,\exp(x+y)$

(%o3) $f(x,y,z):=[f1(x,y,z),f2(x,y,z)]$

```
(%i4) g1(x,y,z):=2*x^2+y^2+3*z+1;      g2(x,y,z):=z^2*sin(x+y);
      g(x,y,z):=[g1(x,y,z),g2(x,y,z)];
```

(%o4) $g1(x,y,z):=2\,x^2+y^2+3\,z+1$

(%o5) $g2(x,y,z):=z^2\sin(x+y)$

(%o6) $g(x,y,z):=[g1(x,y,z),g2(x,y,z)]$

```
(%i7) '((f+g)(x,y,z))=f(x,y,z) + g(x,y,z);
```

(%o7) $(g+f)(x,y,z)=[-\sin(z)+3\,z+y^2+y+3\,x^2+1,\sin(y+x)\,z^2+\%e^{y+x}\,z]$

Producto por un escalar real:

```
(%i8) '((af)(x,y,z))=a*f(x,y,z);
```

(%o8) $af(x,y,z)=[a\,(-\sin(z)+y+x^2),a\;\%e^{y+x}\,z]$

Producto escalar:

```
(%i9) f(x,y,z).g(x,y,z);
```

(%o9) $(3\,z+y^2+2\,x^2+1)(-\sin(z)+y+x^2)+\%e^{y+x}\sin(y+x)\,z^3$

Si se consideran dos funciones vectoriales

$$f : A_1 \subset \mathbb{R}^n \to \mathbb{R}^m, \quad g : A_2 \subset \mathbb{R}^m \to \mathbb{R}^p$$

se define la función compuesta $g \circ f$ mediante:

$$(g \circ f)(x) = g\left(f(x)\right); \quad \mathrm{Dom}(g \circ f) = \left\{x \in A_1,\ f(x) \in A_2\right\}$$

Por ejemplo:

```
(%i1) f1(r,a):=r*cos(theta);        f2(r,a):=r*sin(theta);
      g(x,y):=sqrt(R^2 - x^2 - y^2);
      'F(r,a)=g(f1(r,theta), f2(r,theta));
```

(%o1) $f1(r,a):=r\cos(\theta)$

(%o2) $f2(r,a):=r\sin(\theta)$

(%o3) $g(x,y):=\sqrt{R^2-x^2-y^2}$

(%o4) $F(r,a)=\sqrt{R^2-r^2\sin(\theta)^2-r^2\cos(\theta)^2}$

```
(%i5) trigsimp(%);
```

(%o5) $F(r,a)=\sqrt{R^2-r^2}$

Veamos otro ejemplo:

```
(%i1) g(x,y,z):=x*y^2+exp(z);
      f1(t):=2*t+1;        f2(t):=t^2+2;        f3(t):=sin(t);
      f(t):=[f1(t), f2(t), f3(t)];
```

(%o1) $g(x,y,z):=x\,y^2+\exp(z)$

(%o2) $f1(t):=2\,t+1$

(%o3) $f2(t):=t^2+2$

(%o4) $f3(t):=\sin(t)$

(%o5) $f(t):=[f1(t),f2(t),f3(t)]$

```
(%i6) '(g(f(t)))=g(f1(t),f2(t),f3(t));
```

(%o6) $g(f(t))=\%e^{\sin(t)}+(2\,t+1)(t^2+2)^2$

10-2.- Representación gráfica de funciones reales de dos variables.

Los contenidos de este apartado se desarrollan en el archivo **Tema_10-2.wxm**.

Para representar gráficamente una función real de dos variables hay que:

1) Definir la función
2) Aplicar las opciones del menú "Gráficos 3D", seleccionando las opciones que más interesen para la representación gráfica que se quiera hacer.

El cuadro de diálogo de Gráficos 3D es:



Breve descripción de las opciones del cuadro:

- Expresión: hay que escribir la notación asignada a la función.
- Variable x: hay que especificar el rango de la variable x.
- Variable y: hay que especificar el rango de la variable y.
- Cuadrícula: opción para modificar la graduación de la malla en la que se hace la gráfica.
- Formato: hay cuatro opciones: predeterminado, en línea, gnuplot y openmath. Hacen referencia a cuatro formatos de salida de la gráfica que se verán en los Ejemplos.



- Opciones: hay seis opciones que configuran la forma de la gráfica, tal como se verá en los Ejemplos.



Ejemplo 10.2.1. Se hará la representación gráfica en diferentes formatos de la función

$$f_1(x, y) = 2(x-1)^2 + 2(y-1)^2$$

Definición de la función:

```
(%i1) f2(x,y):=x^2-y^2;
(%o1) f2(x,y):=x^2-y^2
```

Gráfica con el formato "predeterminado" habiendo hecho una rotación de la gráfica obtenida:



Gráfico con el formato "en línea" y diferentes opciones

```
(%i3) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],
      [gnuplot_preamble, "set pm3d at b"])$
```

(%t3)




```
(%i4) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],
      [gnuplot_preamble, "set hidden3d"])$
```

(%t4)




```
(%i5) wxplot3d(f1(x,y), [x,-5,5], [y,-5,5],
      [gnuplot_preamble, "set pm3d at s; unset surf; unset colorbox"])$
```

(%t5)



Ejemplo 10.2.2. Se hará la representación gráfica en diferentes formatos de la función

$$f_1(x, y) = x^2 - y^2$$

Definición de la función:

```
(%i1) f1(x,y):=2*(x-1)^2+2*(y-1)^2;
(%o1) f1(x,y):=2(x-1)^2+2(y-1)^2
```

El formato “predeterminado” abre una ventana que permite ver la gráfica con diferentes ángulos y perspectivas. Se recomienda aplicar esta opción en el archivo de este apartado. Con la opción “en línea” el gráfico queda “enganchado” al propio archivo de cálculos.

Gráfico con el formato “predeterminado” habiendo hecho una rotación de la gráfica obtenida:



Gráfico con el formato “en línea” y diferentes opciones

(%i3) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set pm3d at b"])$

(%t3)



(%i4) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set hidden3d"])$

(%t4)



(%i5) wxplot3d(f2(x,y), [x,-5,5], [y,-5,5],
[gnuplot_preamble, "set pm3d at s; unset surf; unset colorbox"])$

(%t5)



10-3.- Límites de funciones reales de dos variables. Continuidad

Los contenidos de este apartado se desarrollan en el archivo **Tema_10-3.wxm**.

Límite en un punto de una función real de dos variables.
Si $f : A \subset \mathbb{R}^2 \to \mathbb{R}$ y (x_0, y_0) es un punto de acumulación del campo de existencia de la función, se dice que la función tiene límite $L \in \mathbb{R}$ en el punto (x_0, y_0) si cualquiera que sea $\varepsilon > 0$ existe $\delta > 0$ tal que para todo $(x, y) \in \left(B^*\left((x_0, y_0); \delta\right)\right) \cap A$ se cumple $\left| f(x, y) - L \right| < \varepsilon$. Notación: $L = \lim\limits_{(x,y)\to(x_0,y_0)} f(x, y)$.
El límite de una función en un punto, si existe, es único.

Límite en un punto de una función real de dos variables según un subconjunto
Si $f : A \subset \mathbb{R}^2 \to \mathbb{R}$, $(x_0, y_0) \in A'$ y $W \subset \mathbb{R}^2$, se dice que la función tiene límite $L \in \mathbb{R}$ en el punto (x_0, y_0) según el subconjunto *W*, si cualquiera que sea $\varepsilon > 0$ existe $\delta > 0$ tal que para todo $(x, y) \in \left(B^*\left((x_0, y_0); \delta\right)\right) \cap W$ se cumple $\left| f(x, y) - L \right| < \varepsilon$.
Notación:

$$L = \lim_{\substack{(x_0,y_0) \\ W}} f \qquad \text{o be} \qquad L = \lim_{\substack{(x,y)\to(x_0,y_0) \\ (x,y)\in W}} f(x, y)$$

Si el subconjunto *W* es una recta, el límite se llama <u>direccional</u>.

Si existen dos subconjuntos W_1, W_2 tales que el límite respectivo de la función en el punto según cada uno de ellos es diferente, entonces el límite de la función en el punto no existe.

En wxMaxima no hay ninguna instrucción implementada para calcular el límite de funciones de dos ni de más variables. Por lo tanto, hay que aplicar los criterios a este efecto aplicando los cálculos con el programa.

Ejemplo 10.3.1. Se considera la función:

$$f(x, y) = \frac{2x^2 y}{2x^2 + y^2}, \quad \text{si } (x, y) \neq (0, 0)$$

El dominio de esta función es $A = \mathbb{R}^2 - \{(0, 0)\}$ y el punto $(0, 0)$ es un punto de acumulación de *A*. Verifiquemos que $\lim\limits_{(0,0)} f = 0$. En efecto:

$$\left| f(x, y) - 0 \right| = \left| \frac{2x^2 y}{2x^2 + y^2} \right| = \frac{2x^2}{2x^2 + y^2} \left| y \right| \leq \left| y \right| < \varepsilon$$

siempre que $\left|(x,y)-(0,0)\right| = +\sqrt{x^2+y^2} < \delta$, con $\delta \le \varepsilon$.

Ejemplo 10.3.2. Se considera la función:

$$f(x,y) = \frac{x^2}{x^2+y^2}, \quad \text{si } (x,y) \neq (0,0)$$

El dominio de esta función es $A = \mathbb{R}^2 - \{(0,0)\}$ y el punto $(0,0)$ es un punto de acumulación de *A*. Si se considera el subconjunto $W = \{(x,y): y = mx\}$ que es una recta que pasa por el origen, entonces el límite (direccional) de la función según este subconjunto en el punto es:

```
(%i1) f(x,y):=(x^2)/(x^2+y^2);
```

(%o1) $f(x,y) := \frac{x^2}{x^2+y^2}$

```
(%i2) f(x,m*x);
```

(%o2) $\frac{x^2}{m^2 x^2 + x^2}$

```
(%i3) L1=limit( f(x,m*x) , x, 0);
```

(%o3) $L1 = \frac{1}{m^2+1}$

Por lo tanto el límite direccional depende de la recta y en conclusión, el límite de la función en el punto $(0,0)$ no existe.

Límites reiterados. Si $f : A \subset \mathbb{R}^2 \to \mathbb{R}$ y $(x_0, y_0) \in A'$, se llaman <u>límites reiterados</u> de la función en este punto, los límites siguientes:

$$L_{xy} = \lim_{y \to y_0}\left(\lim_{x \to x_0} f(x,y)\right) \;;\; L_{yx} = \lim_{x \to x_0}\left(\lim_{y \to y_0} f(x,y)\right)$$

Cálculo con wxMaxima:

```
(%i1) limit ( limit(f(x,y), y, y0) , x, x0);
```

(%o1) $\lim_{x \to x0} \lim_{y \to y0} f(x,y)$

```
(%i2) limit ( limit(f(x,y), x, x0) , y, y0);
```

$$(\%o2)\quad \lim_{y \to y0}\ \lim_{x \to x0} f(x,y)$$

Propiedades de los límites reiterados:

- Si existe el límite de la función $L = \lim_{(x_0,y_0)} f$, entonces cualquiera de los límites reiterados que exista debe tener este valor.
- Puede que alguno de los límites reiterados o ambos no existan y sí que exista el límite de la función $L = \lim_{(x_0,y_0)} f$.
- Si los límites reiterados existen pero son diferentes, entonces el límite de la función en el punto no existe.
- Si los límites reiterados existen y son iguales, esto no implica que el límite de la función en el punto exista.

Ejemplo 10.3.3. Se considera la función definida por

$$f(x,y) = \frac{x^2 y}{x^4 + y^2}, \quad \text{si } (x,y) \neq (0,0).$$

Entonces se cumple:

```
(%i1) f(x,y):=(x^2 * y)/(x^4+y^2);
```

$$(\%o1)\quad f(x,y):=\frac{x^2\, y}{x^4+y^2}$$

Límites reiterados en el origen:

```
(%i2) 'Lyx(0,0) = limit ( limit(f(x,y), y, 0) , x, 0);
```

$$(\%o2)\quad Lyx(0,0)=0$$

```
(%i3) 'Lxy(0,0)  = limit ( limit(f(x,y), x, 0) , y, 0);
```

$$(\%o3)\quad Lxy(0,0)=0$$

Así, pues, los límites reiterados en el origen existen y son iguales. Sin embargo calculamos el límite direccional según las parábolas de ecuación $y = x^2$:

```
(%i4) 'L[y=x^2](0,0) = limit(f(x,x^2), x, 0);
```

$$(\%o4)\quad L_{y=x^2}(0,0)=\frac{1}{2}$$

Por lo tanto, el límite de la función en el origen no existe.

Ejemplo 10.3.4. Se considera la función definida por:

$$f(x,y)=\begin{cases}(x+y)\sin\left(\dfrac{1}{x}\right)\sin\left(\dfrac{1}{y}\right), & \text{si } (x,y)\neq(0,0)\\ 0, & \text{si } x=0 \text{ o } y=0\end{cases}.$$

Entonces se cumple:

```
(%i1) f(x,y):=(x + y)*sin(1/x)*sin(1/y);
```

$$\text{(\%o1) } f(x,y):=(x+y)\sin\left(\frac{1}{x}\right)\sin\left(\frac{1}{y}\right)$$

Límites reiterados en el origen:

```
(%i2) 'Lyx(0,0) = limit ( limit(f(x,y), y, 0) , x, 0);
(%o2) Lyx(0,0)=ind
```

```
(%i3) 'Lxy(0,0)  = limit ( limit(f(x,y), x, 0) , y, 0);
(%o3) Lxy(0,0)=ind
```

Así, pues, los límites reiterados en el origen no existen. Sin embargo el límite de la función en el origen existe y vale 0 ya que es el producto de una función de límite 0 por una función acotada.

El Teorema de descomposición del dominio.

Sea $f:A\subset\mathbb{R}^2\to\mathbb{R}$ y (x_0,y_0) un punto de acumulación del dominio. Si el dominio de la función se descompone en reunión finita de subconjuntos $A=W_1\cup\cdots\cup W_p$ y se cumple que $L=\lim\limits_{\substack{(x_0,y_0)\\ W_k}} f$, $k=1,2,\ldots,p$ entonces se verifica que $\lim\limits_{(x_0,y_0)} f=L$.

Ejemplo 10.3.5. Para calcular el límite $\lim\limits_{(0,0)}\dfrac{e^{xy}-1}{\sqrt{x^2+y^2}}$ aplicaremos el teorema de descomposición del dominio, de la manera siguiente:

$$\begin{aligned}A&=\mathbb{R}^2-\{(0,0)\}=W_1\cup W_2\cup W_3=\\ &=\{(x,y)\in\mathbb{R}^2: x=0, y\neq 0\}\cup\{(x,y)\in\mathbb{R}^2: x\neq 0, y=0\}\cup\{(x,y)\in\mathbb{R}^2: x\neq 0, y\neq 0\}\end{aligned}$$

verificándose que:

```
(%i1) f(x,y):=(%e^(x*y) - 1) / sqrt(x^2 + y^2) ;
```

$$(\%o1)\ f(x,y):=\frac{\%e^{x\,y}-1}{\sqrt{x^2+y^2}}$$

```
(%i2) 'LW1(0,0) = limit(f(0,y), x, 0);
(%o2) LW1(0,0)=0
```

```
(%i3) 'LW2(0,0) = limit(f(x,0), y, 0);
(%o3) LW2(0,0)=0
```

El límite según el tercer subconjunto se calcula mediante:

$$\lim_{\substack{(0,0)\\ W_3}} \frac{e^{xy}-1}{xy}\cdot\frac{xy}{\sqrt{x^2+y^2}} = \lim_{\substack{(0,0)\\ W_3}} \frac{e^{xy}-1}{xy}\cdot\frac{x}{\sqrt{x^2+y^2}}\,y = 1\cdot \text{fitada}\cdot 0 = 0$$

Por lo tanto, se cumple que $\lim\limits_{(0,0)} \frac{e^{xy}-1}{\sqrt{x^2+y^2}} = 0$.

Continuidad de funciones reales de dos variables

Si $f: A\subset \mathbb{R}^2 \to \mathbb{R}$ y $(x_0,y_0)\in A$, es un punto del dominio que también es de acumulación (es decir es un punto no aislado del dominio), se dice que la función es continua en este punto si se cumple

$$\lim_{(x,y)\to(x_0,y_0)} f(x,y) = f(x_0,y_0)\,.$$

Si en un punto una función no es continua, se dice que es *discontinua*. Se dice que f es continua en $X\subset A$ si la función es continua en todos los puntos de este subconjunto. El mayor subconjunto del dominio en que una función es continua se llama campo de continuidad de la función.

10-4.- Derivadas parciales. Derivadas direccionales. Derivadas de orden superior.

Los contenidos de este apartado se desarrollan en el archivo **Tema_10-4.wxm**.

10.4.1. Derivadas parciales de funciones reales de dos variables

Consideremos una función $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ y un punto $a = (a_1, a_2) \in \overset{o}{A}$. Se definen las <u>derivadas parciales</u> de f en el punto a mediante:

$$D_x f(a) = \lim_{t \to 0} \frac{f(a_1 + t, a_2) - f(a_1, a_2)}{t} \quad ; \quad D_y f(a) = \lim_{t \to 0} \frac{f(a_1, a_2 + t) - f(a_1, a_2)}{t}$$

Notación:

$$D_x f(a) \equiv D_1 f(a) \equiv \frac{\partial f}{\partial x}(a) \quad ; \quad D_y f(a) \equiv D_2 f(a) \equiv \frac{\partial f}{\partial y}(a)$$

Una función $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ se dice <u>derivable parcialmente</u> en $a \in \overset{o}{A}$, si en este punto existen las dos derivadas parciales de la función.
Una función $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ se dice <u>derivable parcialmente</u> en un subconjunto del dominio si existen las dos derivadas parciales de la función en todos los puntos de este conjunto. Las funciones que asignan a cada punto la derivada parcial correspondiente se llaman <u>funciones derivadas parciales</u> de *f*.

Veamos como se calculan las derivadas parciales con wxMaxima.

Ejemplo 10.4.1. Si se considera la función definida por $f(x, y) = 4x^2 + 3y^2$, se trata de calcular las funciones derivadas parciales y las derivadas parciales en el punto $a = (1, 2)$.

Primero hay que definir la función:

```
(%i1) f(x,y):=4*x^2 + 3*y^2;
```
(%o1) $f(x,y):=4\,x^2+3\,y^2$

A continuación hay que aplicar las instrucciones de cálculo de derivadas, análogas al caso de funciones de una variable, pero en este caso se ha de especificar la variable respecto a la que se deriva. Se puede obtener la expresión de la función derivada parcial correspondiente con la instrucción "define":

```
(%i2) define (Dxf(x,y) , diff(f(x,y), x) );
      define (Dyf(x,y) , diff(f(x,y), y) );
```

(%o2) $Dxf(x,y):=8\,x$

(%o3) $Dyf(x,y):=6\,y$

Y finalmente se puede calcular el valor de las funciones derivadas parciales en el punto correspondiente:

```
(%i4) 'Dxf(1,2)=Dxf(1,2);
      'Dyf(1,2)=Dyf(1,2);
```

(%o4) $Dxf(1,2)=8$

(%o5) $Dyf(1,2)=12$

En algunos casos el cálculo de la derivada parcial en un punto no se puede llevar a cabo de una forma tan sencilla y hay que recurrir a la definición para verificar si la función es o no derivable parcialmente en el punto, tal como se muestra en el Ejemplo que sigue.

Ejemplo 10.4.2. Se considera la función definida por

$$f(x,y)=\begin{cases} \dfrac{x^3+y^3}{x^2+y^2}, & \text{si } (x,y)\neq(0,0) \\ 0, & \text{si } (x,y)=(0,0) \end{cases}.$$

Se trata de calcular las funciones derivadas parciales, las derivadas parciales en un punto cualquiera $(x_0,y_0)\neq(0,0)$ y las derivadas parciales en el punto $(0,0)$. Comencemos calculando las derivadas parciales de esta función:

```
(%i1) f(x,y):=(x^3 + y^3)/(x^2 + y^2);
```

(%o1) $f(x,y):=\dfrac{x^3+y^3}{x^2+y^2}$

```
(%i2) define (Dxf(x,y) , diff(f(x,y), x) );
      define (Dyf(x,y) , diff(f(x,y), y) );
```

(%o2) $Dxf(x,y):=\dfrac{3\,x^2}{y^2+x^2}-\dfrac{2\,x\,(y^3+x^3)}{(y^2+x^2)^2}$

(%o3) $Dyf(x,y):=\dfrac{3\,y^2}{y^2+x^2}-\dfrac{2\,y\,(y^3+x^3)}{(y^2+x^2)^2}$

Ahora ya podemos calcular las derivadas parciales en un punto cualquiera $(x_0,y_0)\neq(0,0)$:

(%i4) 'Dxf(x0,y0) = Dxf(x0,y0); 'Dyf(x0,y0) = Dyf(x0,y0);

(%o4) $Dxf(x0,y0)=\frac{3\,x0^2}{y0^2+x0^2}-\frac{2\,x0\,(y0^3+x0^3)}{(y0^2+x0^2)^2}$

(%o5) $Dyf(x0,y0)=\frac{3\,y0^2}{y0^2+x0^2}-\frac{2\,y0\,(y0^3+x0^3)}{(y0^2+x0^2)^2}$

Las derivadas en el origen no se pueden calcular aplicando las funciones derivadas parciales en este punto. En efecto:

```
(%i6) 'Dxf(0,0)=Dxf(0,0);     'Dyf(0,0)=Dyf(0,0);
Division by 0
  #0: Dxf(x=0,y=0)
  -- an error. To debug this try: debugmode(true);
  Division by 0
  #0: Dyf(x=0,y=0)
  -- an error. To debug this try: debugmode(true);
```

Por lo tanto se ha de aplicar la definición de derivada parcial:

```
(%i8) 'Dxf(0,0) = limit( (f(t,0) - 0)/(t) , t, 0);
(%o8) Dxf(0,0)=1
```

```
(%i9) 'Dyf(0,0) =limit( (f(0,t) - 0)/(t) , t, 0);
(%o9) Dyf(0,0)=1
```

Así pues la función es derivable parcialmente en el origen y se ha calculado el valor de cada una de estas derivadas parciales.

10.4.2 Derivadas direccionales de funciones reales de dos variables

Consideremos una función $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ y un punto $a \in \overset{o}{A}$. Se define la derivada de f en este punto en la dirección del vector unitario $u \in \mathbb{R}^2$ como el límite, si existe:

$$D_u f(a) = \lim_{t\to 0}\frac{f(a+tu)-f(a)}{t}$$

Notación: $D_u f(a) \equiv \frac{\partial f}{\partial u}(a)$.

Ejemplo 10.4.3. Se considera la función definida por $f(x,y)=x+2y$; se quiere calcular la derivada direccional según el vector $v=(1,1)$ en el punto $a=(3,2)$.

En primer lugar se observa que el vector dado no es unitario y, por lo tanto, hay que normalizarlo:

```
(%i1) 'u=u:[1,1];     modul:sqrt(1^2+1^2);        'u[1]=u/modul;
```

(%o1) $u=[1,1]$

(%o2) $\sqrt{2}$

(%o3) $u_1=[\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}}]$

Ahora definimos la función:

```
(%i4) f(x,y):=x+2*y;
```

(%o4) $f(x,y):=x+2\,y$

Y procedemos al cálculo de la derivada direccional:

```
(%i5) 'Duf(3,2) = limit(( f(3+t/sqrt(2),2+t/sqrt(2)) - f(3,2))/t, t, 0);
```

(%o5) $Duf(3,2)=\sqrt{2}+\frac{1}{\sqrt{2}}$

10.4.3. Derivadas de orden superior de funciones reales de dos variables

Consideremos una función $f:A\subset\mathbb{R}^2\to\mathbb{R}$. Si la función es derivable parcialmente en un subconjunto $\Omega\subset A$, existen las derivadas parciales $D_1 f, D_2 f$ definidas en Ω. Si estas funciones son derivables parcialmente, sus derivadas parciales se llaman <u>derivadas parciales de segundo orden</u>.

Para cada $a=(a_1,a_2)\in\Omega$ se definen así:

$$D_{xx}f(a)=D_x(D_x f)(a)=\lim_{t\to 0}\frac{D_x f(a_1+t,a_2)-D_x f(a_1,a_2)}{t}$$

$$D_{xy}f(a)=D_y(D_x f)(a)=\lim_{t\to 0}\frac{D_x f(a_1,a_2+t)-D_x f(a_1,a_2)}{t}$$

$$D_{yx}f(a)=D_x(D_y f)(a)=\lim_{t\to 0}\frac{D_y f(a_1+t,a_2)-D_y f(a_1,a_2)}{t}$$

$$D_{yy}f(a)=D_y(D_y f)(a)=\lim_{t\to 0}\frac{D_y f(a_1,a_2+t)-D_y f(a_1,a_2)}{t}$$

Notación:

$$D_{xx} f(a) \equiv D_{11} f(a) \equiv \frac{\partial^2 f}{\partial x^2}(a) \quad ; \quad D_{xy} f(a) \equiv D_{12} f(a) \equiv \frac{\partial^2 f}{\partial x \partial y}(a)$$

$$D_{yx} f(a) \equiv D_{21} f(a) \equiv \frac{\partial^2 f}{\partial y \partial x}(a) \quad ; \quad D_{yy} f(a) \equiv D_{22} f(a) \equiv \frac{\partial^2 f}{\partial y^2}(a)$$

Metodología de cálculo: dado que se trata de derivar parcialmente, se aplica la misma metodología que se ha comentado para las derivadas parciales de primer orden.

Las derivadas $D_{xy} f(x,y)$ y $D_{yx} f(x,y)$, se llaman derivadas cruzadas (o mixtas). Por estas derivadas hay un teorema importante.

Teorema (Schwarz). Si una función $f : A \subset \mathbb{R}^2 \to \mathbb{R}$ es tal que en un entorno del punto $(a,b) \in A$ existen las derivadas parciales $D_x f(x,y)$, $D_y f(x,y)$, $D_{xy} f(x,y)$ y esta derivada parcial de segundo orden es continua en el punto $(a,b) \in A$, entonces existe la derivada de segundo orden $D_{yx} f(a,b)$ y se cumple la igualdad $D_{yx} f(a,b) = D_{xy} f(a,b)$.

La matriz hessiana. Si una función $f : A \subset \mathbb{R}^2 \to \mathbb{R}$ es derivable dos veces en un punto $(a,b) \in A$, se llama matriz hessiana de la función en este punto, la matriz cuadrada

$$Hf(a,b) = \begin{pmatrix} D_{xx} f(a,b) & D_{xy} f(a,b) \\ D_{yx} f(a,b) & D_{yy} f(a,b) \end{pmatrix}$$

Ejemplo 10.4.4. Se considera la función definida por:

$$f(x,y) = 2x^3 y - xy^2 + e^{xy}.$$

Derivadas parciales de primer orden:

```
(%i1) f(x,y):=2*x^3*y - x*y^2 + exp(x*y);
(%o1) f(x,y):=2 x^3 y-x y^2+exp(x y)
```

```
(%i2) define (Dxf(x,y) , diff(f(x,y), x) );
      define (Dyf(x,y) , diff(f(x,y), y) );
(%o2) Dxf(x,y):=y %e^(x y)-y^2+6 x^2 y
(%o3) Dyf(x,y):=x %e^(x y)-2 x y+2 x^3
```

Derivadas parciales de segundo orden:

```
(%i4) define (Dxxf(x,y) , diff(Dxf(x,y), x) );
      define (Dxyf(x,y) , diff(Dxf(x,y), y) );
```

(%o4) $Dxxf(x,y):=y^2\ \%e^{x\,y}+12\,x\,y$

(%o5) $Dxyf(x,y):=x\,y\ \%e^{x\,y}+\%e^{x\,y}-2\,y+6$

```
(%i6) define (Dyxf(x,y) , diff(Dyf(x,y), x) );
      define (Dyyf(x,y) , diff(Dyf(x,y), y) );
```

(%o6) $Dyxf(x,y):=x\,y\ \%e^{x\,y}+\%e^{x\,y}-2\,y+6\,x^2$

(%o7) $Dyyf(x,y):=x^2\ \%e^{x\,y}-2\,x$

Para calcular la matriz hessiana se puede construir una matriz con las derivadas de segundo orden calculadas, o bien utilizar la instrucción "hessian" de wxMaxima que permite obtener esta matriz directamente:

```
(%i8) Hf(x,y) = matrix ( [Dxxf(x,y), Dxyf(x,y)] , [Dyxf(x,y), Dyyf(x,y)] ) ;
```

(%o8) $Hf(x,y)=\begin{bmatrix} y^2\ \%e^{x\,y}+12\,x\,y & x\,y\ \%e^{x\,y}+\%e^{x\,y}-2\,y+6\,x^2 \\ x\,y\ \%e^{x\,y}+\%e^{x\,y}-2\,y+6\,x^2 & x^2\ \%e^{x\,y}-2\,x \end{bmatrix}$

```
(%i9) Hf(x,y) = Hf: hessian (f(x,y) , [x,y]) ;
```

(%o9) $Hf(x,y)=\begin{bmatrix} y^2\ \%e^{x\,y}+12\,x\,y & x\,y\ \%e^{x\,y}+\%e^{x\,y}-2\,y+6\,x^2 \\ x\,y\ \%e^{x\,y}+\%e^{x\,y}-2\,y+6\,x^2 & x^2\ \%e^{x\,y}-2\,x \end{bmatrix}$

Para calcular la matriz hessiana en un punto, la sintaxis de wxMaxima es la siguiente:

```
(%i10) 'Hf(1,0) = Hf10:subst( [x=1, y=0] , Hf);
```

(%o10) $Hf(1,0)=\begin{bmatrix} 0 & 7 \\ 7 & -1 \end{bmatrix}$

10.4.4. Derivadas parciales de funciones reales de varias variables

Para funciones reales de más de dos variables la metodología de cálculo de derivadas parciales de primer orden, de derivadas direccionales y derivadas parciales de orden superior es análoga al caso de dos variables. Veremos la sintaxis mediante un ejemplo.

Ejemplo 10.4.5. Se considera la función definida por:

$$f(x,y,z)=4x^2y^2+3xe^{z^2}\ .$$

Definición de la función:

```
(%i1) f(x,y,z):=4*x^2*y^2 + 3*x*exp(z^2);
```

(%o1) $f(x,y,z):=4\,x^2\,y^2+3\,x\,\exp(z^2)$

Derivadas parciales de primer orden:

```
(%i2) define (Dxf(x,y,z) , diff(f(x,y,z), x) );
      define (Dyf(x,y,z) , diff(f(x,y,z), y) );
      define (Dzf(x,y,z) , diff(f(x,y,z), z) );
```

(%o2) $Dxf(x,y,z):=3\,\%e^{z^2}+8\,x\,y^2$

(%o3) $Dyf(x,y,z):=8\,x^2\,y$

(%o4) $Dzf(x,y,z):=6\,x\,z\,\%e^{z^2}$

Derivadas parciales en un punto:

```
(%i5) 'Dxf(1,2,-1) = Dxf(1,2,-1);
      'Dyf(1,2,-1) = Dyf(1,2,-1);
      'Dzf(1,2,-1) = Dzf(1,2,-1);
```

(%o5) $Dxf(1,2,-1)=3\,\%e+32$

(%o6) $Dyf(1,2,-1)=16$

(%o7) $Dzf(1,2,-1)=-6\,\%e$

Derivadas parciales de segundo orden:

```
(%i8) define (Dxxf(x,y,z) , diff(Dxf(x,y,z), x) );
      define (Dxyf(x,y,z) , diff(Dxf(x,y,z), y) );
      define (Dxzf(x,y,z) , diff(Dxf(x,y,z), z) );
```

(%o8) $Dxxf(x,y,z):=8\,y^2$

(%o9) $Dxyf(x,y,z):=16\,x\,y$

(%o10) $Dxzf(x,y,z):=6\,z\,\%e^{z^2}$

```
(%i11) define (Dyxf(x,y,z) , diff(Dyf(x,y,z), x) );
       define (Dyyf(x,y,z) , diff(Dyf(x,y,z), y) );
       define (Dyzf(x,y,z) , diff(Dyf(x,y,z), z) );
```

(%o11) $Dyxf(x,y,z):=16\,x\,y$

(%o12) $Dyyf(x,y,z):=8\,x^2$

(%o13) $Dyzf(x,y,z):=0$

```
(%i14) define (Dzxf(x,y,z) , diff(Dzf(x,y,z), x) );
       define (Dzyf(x,y,z) , diff(Dzf(x,y,z), y) );
       define (Dzzf(x,y,z) , diff(Dzf(x,y,z), z) );
```

(%o14) $\mathrm{Dzxf}(x,y,z):=6\,z\,\%e^{z^2}$

(%o15) $\mathrm{Dzyf}(x,y,z):=0$

(%o16) $\mathrm{Dzzf}(x,y,z):=12\,x\,z^2\,\%e^{z^2}+6\,x\,\%e^{z^2}$

Matriz hessiana:

```
(%i17) Hf(x,y,z) = matrix( [ Dxxf(x,y,z), Dxyf(x,y,z), Dxzf(x,y,z)] ,
                           [ Dyxf(x,y,z), Dyyf(x,y,z), Dyzf(x,y,z)] ,
                           [ Dzxf(x,y,z), Dzyf(x,y,z), Dzzf(x,y,z)]     );
```

(%o17) $\mathrm{Hf}(x,y,z)=\begin{bmatrix} 8\,y^2 & 16\,x\,y & 6\,z\,\%e^{z^2} \\ 16\,x\,y & 8\,x^2 & 0 \\ 6\,z\,\%e^{z^2} & 0 & 12\,x\,z^2\,\%e^{z^2}+6\,x\,\%e^{z^2} \end{bmatrix}$

```
(%i18) Hf(x,y) = Hf: hessian (f(x,y,z) , [x,y,z]) ;
```

(%o18) $\mathrm{Hf}(x,y)=\begin{bmatrix} 8\,y^2 & 16\,x\,y & 6\,z\,\%e^{z^2} \\ 16\,x\,y & 8\,x^2 & 0 \\ 6\,z\,\%e^{z^2} & 0 & 12\,x\,z^2\,\%e^{z^2}+6\,x\,\%e^{z^2} \end{bmatrix}$

Matriz hessiana en un punto:

```
(%i19) 'Hf(1,2,-1) = Hf10:subst( [x=1, y=2, z=-1] , Hf);
```

(%o19) $\mathrm{Hf}(1,2,-1)=\begin{bmatrix} 32 & 32 & -6\,\%e \\ 32 & 8 & 0 \\ -6\,\%e & 0 & 18\,\%e \end{bmatrix}$

10-5.- Matriz jacobiana. Diferenciabilidad. Regla de la cadena.

Los contenidos de este apartado se desarrollan en el archivo **Tema_10-5.wxm**.

10.5.1 La matriz jacobiana.

Si la función $f : A \subset \mathbb{R}^2 \to \mathbb{R}$ es derivable parcialmente en un punto $a \in \overset{o}{A}$ se llama matriz jacobiana de f en este punto la matriz fila de dos columnas definida por:

$$Jf(a) = \begin{pmatrix} D_x f(a) & D_y f(a) \end{pmatrix}$$

Ejemplo 10.5.1. Se considera la función definida por

$$f(x, y) = 2x^2 y - 3xy^3$$

La sintaxis para el cálculo de la matriz jacobiana en un punto cualquiera y en un punto concreto es la siguiente:

```
(%i1) f(x,y):=2*x^2*y-3*x*y^3;
(%o1) f(x,y):=2 x^2 y-3 x y^3

(%i2) 'Jf(x,y) = Jf: jacobian([f(x,y)],[x,y]);
(%o2) Jf(x,y)=[4 x y-3 y^3   2 x^2-9 x y^2]

(%i3) 'Jf(1,2) = subst([x=1,y=2], Jf);
(%o3) Jf(1,2)=[-16   -34]
```

Si se considera una función real de varias variables $f : A \subset \mathbb{R}^n \to \mathbb{R}$ tal que es derivable parcialmente en un punto $a \in \overset{o}{A}$ se llama matriz jacobiana de la función en este punto la matriz fila de n columnas definida por:

$$Jf(a) = \begin{pmatrix} D_1 f(a) & D_2 f(a) & \cdots & D_n f(a) \end{pmatrix}$$

Ejemplo 10.5.2. Se considera la función definida por

$$f(x, y, z, t) = 2x^2 y - 3zy^3 + 2z^2 \sin(t)$$

La sintaxis para el cálculo de la matriz jacobiana en un punto cualquiera y en un punto concreto es la siguiente:

```
(%i1) f(x,y,z,t):=2*x^2*y-3*z*y^3+2*z^2*sin(t);
(%o1) f(x,y,z,t):=2 x^2 y-3 z y^3+2 z^2 sin(t)

(%i2) 'Jf(x,y,z,t) = Jf: jacobian([f(x,y,z,t)],[x,y,z,t]);
(%o2) Jf(x,y,z,t)=[4 x y  2 x^2-9 y^2 z  4 sin(t) z-3 y^3  2 cos(t) z^2]

(%i3) 'Jf(1,2,0,1) = subst([x=1,y=2,z=0,t=1], Jf);
(%o3) Jf(1,2,0,1)=[8  2  -24  0]
```

Si se considera una función vectorial de varias variables $f : A \subset \mathbb{R}^n \to \mathbb{R}^m$ tal que las funciones componentes son derivables parcialmente en un punto $a \in \overset{o}{A}$, se llama matriz jacobiana de f en el punto *a*, la matriz de *m* filas y *n* columnas y coeficientes reales:

$$Jf(a) = \left(D_k f_i(a)\right)_{\substack{1\le i\le m\\ 1\le k\le n}} = \begin{pmatrix} D_1 f_1(a) & D_2 f_1(a) & \cdots & D_n f_1(a) \\ D_1 f_2(a) & D_2 f_2(a) & \cdots & D_n f_2(a) \\ \cdots & & \cdots & \\ D_1 f_m(a) & D_2 f_m(a) & \vdots & D_n f_m(a) \end{pmatrix}$$

Si $m = n$ la matriz jacobiana es cuadrada y su determinante se llama <u>jacobiano</u> de la función en el punto *a*.

Ejemplo 10.5.3. Se considera la función definida por

$$f(x,y) = \left(2x^2y - 3xy^3, e^{x+y}, xy - 2x^2y\right)$$

La sintaxis para el cálculo de la matriz jacobiana en un punto cualquiera y en un punto concreto es la siguiente:

```
(%i1) f1(x,y):=2*x^2*y-3*x*y^3;     f2(x,y):=exp(x+y);     f3(x,y):=x*y-2*x^2*y;
      f(x,y):= [f1(x,y) , f2(x,y) , f3(x,y)];
(%o1) f1(x,y):=2 x^2 y-3 x y^3
(%o2) f2(x,y):=exp(x+y)
(%o3) f3(x,y):=x y-2 x^2 y
(%o4) f(x,y):=[f1(x,y),f2(x,y),f3(x,y)]
```

```
(%i5) 'Jf(x,y) = Jf: jacobian( f(x,y), [x,y]);
```

$$(\%o5)\ Jf(x,y)=\begin{bmatrix} 4\,x\,y-3\,y^3 & 2\,x^2-9\,x\,y^2 \\ \%e^{y+x} & \%e^{y+x} \\ y-4\,x\,y & x-2\,x^2 \end{bmatrix}$$

```
(%i5) 'Jf(1,-1) = subst([x=1,y=-1], Jf);
```

$$(\%o5)\ Jf(1,-1)=\begin{bmatrix} -1 & -7 \\ 1 & 1 \\ 3 & -1 \end{bmatrix}$$

10.5.2. Diferenciabilidad

Consideremos $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ una función y un punto $a=(a_1,a_2)\in \overset{\circ}{A}$. Si existe la matriz jacobiana de *f* en el punto a, $Jf(a)=\left(D_x f(a) \quad D_y f(a)\right)$, se llama <u>diferencial</u> de *f* en el punto *a* la aplicación lineal $df_a: \mathbb{R}^2 \to \mathbb{R}$ tal que su matriz en las bases canónicas es $Jf(a)$, es decir, tal que $\left[df_a\right]_{B_c B_c^{'}} = Jf(a)$. En virtud de esta definición, para un vector $h=(h_1,h_2)\in\mathbb{R}^2$ se cumple:

$$df_a(h) = Jf(a)\left[h\right] = \left(D_x f(a) \quad D_y f(a)\right)\begin{pmatrix} h_1 \\ h_2 \end{pmatrix} = h_1 D_x f(a) + h_2 D_y f(a)$$

Se dice que $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ es <u>diferenciable en el punto</u> $a=(a_1,a_2)\in \overset{\circ}{A}$, si la función df_a cumple la condición siguiente, llamada <u>condición de diferenciabilidad</u>:

$$\lim_{h\to 0} \frac{f(a+h)-f(a)-df_a(h)}{|h|} = 0$$

es decir:

$$\lim_{h\to 0} \frac{f(a_1+h_1,a_2+h_2)-f(a_1,a_2)-\left(h_1 D_x f(a)+h_2 D_y f(a)\right)}{\sqrt{h_1^2+h_2^2}} = 0$$

La condición de diferenciabilidad se interpreta así: la función diferencial df_a aproxima localmente los incrementos de la función. Además es la única aplicación lineal que lo verifica.

La existencia de las derivadas parciales es condición necesaria para la diferenciabilidad, pero no es suficiente.

Condición suficiente de diferenciabilidad. Si $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ es derivable parcialmente y las derivadas de primer orden son funciones continuas en el punto $a \in \overset{o}{A}$ entonces la función es diferenciable en este punto.

Propiedades de las funciones diferenciables en un punto. Si $f: A \subset \mathbb{R}^2 \to \mathbb{R}$ y $a \in \overset{o}{A}$, se cumplen las propiedades siguientes:

(1) Si la función es diferenciable en el punto, entonces es continua en este punto (el recíproco no es necesariamente cierto).

(2) Si la función es diferenciable en el punto, entonces es derivable en cualquier dirección y para todo vector unitario $u \in \mathbb{R}^2$, se cumple $D_u f(a) = Jf(a)[u]$.

(3) El recíproco de la propiedad anterior no es cierto, es decir, una función puede tener derivada en cualquier dirección en un punto y no ser diferenciable en este punto.

Ejemplo 10.5.4. Se considera la función definida por

$$f(x,y) = x^2 + 2y^2$$

Esta claro que esta función es diferenciable ya que cumple la condición suficiente de diferenciabilidad. Para calcular la derivada en el punto (3,2) y en la dirección definida por el vector $u = (1,1) \in \mathbb{R}^2$ en primer lugar se normalizará el vector, después se calculará la matriz jacobiana de la función en el punto y finalmente se llevará a cabo el producto de la matriz jacobiana por la matriz columna asociada al vector.

Veamos los cálculos con wxMaxima:

(%i1) f(x,y):=x^2+2*y^2;

(%o1) $f(x,y):=x^2+2\,y^2$

(%i2) 'u=u:[1,1]; modul:sqrt(1^2+1^2); 'u1=u/modul;

(%o2) $u=[1,1]$

(%o3) $\sqrt{2}$

(%o4) $u1=[\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}}]$

```
(%i5) 'u1=u1 : matrix( [1/sqrt(2)] , [1/sqrt(2)] );
```

$$(\%o5)\ u1=\begin{bmatrix}\frac{1}{\sqrt{2}}\\ \frac{1}{\sqrt{2}}\end{bmatrix}$$

```
(%i6) 'Jf(x,y)=Jfxy : jacobian([f(x,y)], [x,y]);
```

$$(\%o6)\ Jf(x,y)=\begin{bmatrix}2x & 4y\end{bmatrix}$$

```
(%i7) 'Jf(3,2) = JfP : subst([x=3,y=2], Jfxy);
```

$$(\%o7)\ Jf(3,2)=\begin{bmatrix}6 & 8\end{bmatrix}$$

```
(%i8) 'Duf(3,2)=JfP.u1;
```

$$(\%o8)\ Duf(3,2)=2^{5/2}+3\sqrt{2}$$

10.5.3 La regla de la cadena

Consideremos dos funciones $f : A \subset \mathbb{R}^n \to \mathbb{R}^m$, $g : B \subset \mathbb{R}^m \to \mathbb{R}^p$ tales que:

1) la función f es diferenciable en $a \in \overset{o}{A}$,

2) la función g es diferenciable en $b = f(a) \in \overset{o}{B}$,

3) se cumple $f(A) \subset B$.

En estas condiciones se puede considerar la función compuesta

$$F = g \circ f : A \subset \mathbb{R}^n \to \mathbb{R}^p,$$

la cual cumple las propiedades siguientes:

1) La función F es diferenciable en el punto $a \in \overset{o}{A}$;

2) Se cumplen las igualdades:

$$dF_a = dg_b \circ df_a; \quad JF(a) = Jg(b)Jf(a)$$

Ejemplo 10.5.5. Se consideran las funciones diferenciables:

$$f(x,y) = (x^2 + y, xy^2, x + y^2),\ g(x,y,z) = (e^{xy}, y + z).$$

Se trata de calcular la matriz jacobiana de la función compuesta $F = g \circ f$ en un punto cualquiera y calcular $JF(1,-1)$.

En primer lugar definimos las funciones:

```
(%i1) f1(x,y):=x^2 + y$     f2(x,y):=x * y^2$     f3(x,y):=x + y^2$
      f(x,y):= [f1(x,y) , f2(x,y) , f3(x,y)];
(%o4) f(x,y):=[f1(x,y),f2(x,y),f3(x,y)]
```

```
(%i5) g1(x,y,z):=%e^(x*y)$     g2(x,y,z):= y + z$
      g(x,y,z):= [g1(x,y,z) , g2(x,y,z) ];
(%o7) g(x,y,z):=[g1(x,y,z),g2(x,y,z)]
```

Cálculo de las respectivas matrices jacobianas:

```
(%i8) 'Jf(x,y)=Jfxy : jacobian(f(x,y), [x,y]);
```

$$(\%o8)\ Jf(x,y)=\begin{bmatrix} 2\,x & 1 \\ y^2 & 2\,x\,y \\ 1 & 2\,y \end{bmatrix}$$

```
(%i9) 'Jg(x,y,z)=Jgxyz : jacobian(g(x,y,z), [x,y,z]);
```

$$(\%o9)\ Jg(x,y,z)=\begin{bmatrix} y\ \%e^{x\,y} & x\ \%e^{x\,y} & 0 \\ 0 & 1 & 1 \end{bmatrix}$$

La matriz jacobiana de la función compuesta en un punto cualquiera se calcula como el producto de matrices:

```
(%i10) 'JF(x,y) = Jgxyz.Jfxy;
```

$$(\%o10)\ JF(x,y)=\begin{bmatrix} x\,y^2\ \%e^{x\,y}+2\,x\,y\ \%e^{x\,y} & 2\,x^2\,y\ \%e^{x\,y}+y\ \%e^{x\,y} \\ y^2+1 & 2\,x\,y+2\,y \end{bmatrix}$$

Finalmente, en el punto (1, -1):

```
(%i11) 'Jf(1,-1) = Jfa : subst([x=1,y=-1], Jfxy);
```

$$(\%o11)\ Jf(1,-1)=\begin{bmatrix} 2 & 1 \\ 1 & -2 \\ 1 & -2 \end{bmatrix}$$

10.5.4 Derivación de funciones compuestas.

En el caso general de la regla de la cadena, si se cumple $m = 1$, la función compuesta es una función real. Este hecho permite el cálculo de las derivadas de la función compuesta aplicando la regla de la cadena. Veremos a continuación tres ejemplos que ilustran los posibles casos.

En primer lugar el caso $n = 1$, $m > 1$, que es la composición de una función vectorial de variable real y una función real de variable vectorial.

Ejemplo 10.5.6. Se consideran las funciones definidas por

$$f(t) = (2t+1, t^2+2, \sin t) \quad , \quad g(x,y,x) = xy^2 + e^z$$

En primer lugar definimos las funciones y la función compuesta:

```
(%i1) f1(t):=2*t + 1;     f2(t):= t^2+2;      f3(t):=sin(t);
      g(x,y,z):=x*y^2 + exp(z);
```

$(\%o1)\ f1(t) := 2\,t+1$

$(\%o2)\ f2(t) := t^2+2$

$(\%o3)\ f3(t) := \sin(t)$

$(\%o4)\ g(x,y,z) := x\,y^2+\exp(z)$

```
(%i5) F(t)=g(f1(t), f2(t), f3(t));
```

$(\%o5)\ F(t) = \%e^{\sin(t)} + (2\,t+1)\,(t^2+2)^2$

Ahora podemos calcular la derivada de la función compuesta:

```
(%i6) DF(t)=diff(g(f1(t), f2(t), f3(t)) , t);
```

$(\%o6)\ DF(t) = \%e^{\sin(t)}\cos(t) + 2\,(t^2+2)^2 + 4\,t\,(2\,t+1)\,(t^2+2)$

```
(%i7) expand(%);
```

$(\%o7)\ DF(t) = \%e^{\sin(t)}\cos(t) + 10\,t^4 + 4\,t^3 + 24\,t^2 + 8\,t + 8$

En segundo lugar veremos un ejemplo del caso $n > 1$, $m = 1$, que es la composición de una función real de variable vectorial y una función real de variable real.

Ejemplo 10.5.7. Se consideran las funciones definidas por

$$f(x,y,z) = \sqrt{xy^2 + e^z} \quad , \quad g(t) = \sin t$$

En primer lugar definimos las funciones y la función compuesta:

```
(%i1) g(t):=sin(t);
      f(x,y,z):=sqrt( x*y^2 + exp(z)) ;
```

(%o1) $g(t):=\sin(t)$

(%o2) $f(x,y,z):=\sqrt{x\,y^2+\exp(z)}$

```
(%i3) F(x,y,z)=g(f(x,y,z));
```

(%o3) $F(x,y,z)=\sin(\sqrt{\%e^z+x\,y^2})$

Las derivadas parciales de la función compuesta se calculan de la manera siguiente:

```
(%i4) DxF(x,y,z)=diff(g(f(x,y,z)) , x);
```

(%o4) $DxF(x,y,z)=\dfrac{y^2\cos(\sqrt{\%e^z+x\,y^2})}{2\sqrt{\%e^z+x\,y^2}}$

```
(%i5) DyF(x,y,z)=diff(g(f(x,y,z)) , y);
```

(%o5) $DyF(x,y,z)=\dfrac{x\,y\cos(\sqrt{\%e^z+x\,y^2})}{\sqrt{\%e^z+x\,y^2}}$

```
(%i6) DzF(x,y,z)=diff(g(f(x,y,z)) , z);
```

(%o6) $DzF(x,y,z)=\dfrac{\%e^z\cos(\sqrt{\%e^z+x\,y^2})}{2\sqrt{\%e^z+x\,y^2}}$

En tercer lugar veremos un ejemplo del caso $n > 1$, $m > 1$, que es la composición de una función vectorial de variable vectorial y una función real de variable vectorial.

Ejemplo 10.5.8. Se consideran las funciones definidas por

$$f(x,y)=(x+y,x-y,2xy) \quad , \quad g(u,v,w)=\log(u^2+v^2-2w)$$

En primer lugar definimos las funciones y la función compuesta:

```
(%i1) f1(x,y):=x+y;     f2(x,y):= x-y;     f3(x,y):=2*x*y;
      g(u,v,w):=log(u^2 + v^2 - 2*w) ;
```

(%o1) $f1(x,y):=x+y$

(%o2) $f2(x,y):=x-y$

(%o3) $f3(x,y):=2\,x\,y$

(%o4) $g(u,v,w):=\log(u^2+v^2+(-2)\,w)$

(%i5) F(x,y)=g(f1(x,y), f2(x,y), f3(x,y));

(%o5) $F(x,y)=\log((y+x)^2-4\,x\,y+(x-y)^2)$

Las derivadas parciales de la función compuesta se calculan de la manera siguiente:

(%i6) DxF(x,y)=diff(g(f1(x,y), f2(x,y), f3(x,y)) , x);

(%o6) $DxF(x,y)=\dfrac{2\,(y+x)-4\,y+2\,(x-y)}{(y+x)^2-4\,x\,y+(x-y)^2}$

(%i7) ratsimp(%);

(%o7) $DxF(x,y)=-\dfrac{2}{y-x}$

(%i8) DyF(x,y)=diff(g(f1(x,y), f2(x,y), f3(x,y)) , y);

(%o8) $DyF(x,y)=\dfrac{2\,(y+x)-2\,(x-y)-4\,x}{(y+x)^2-4\,x\,y+(x-y)^2}$

(%i9) ratsimp(%);

(%o9) $DyF(x,y)=\dfrac{2}{y-x}$